

# SQL\_PREPARE

## Akcia SQL\_PREPARE

**Funkcia**

Akcia pripraví vykonávanie SQL príkazu **SELECT**.

**Deklarácia**

```
SQL_PREPARE handleIdent_Int, retCodeIdent_Int, selectStringExpr BIND
_locVar1, _locVar2, ...

SQL_PREPARE handleIdent_Int, retCodeIdent_Int, selectStringExpr BIND
_locVarRowIdent

SQL_PREPARE handleIdent_Int, retCodeIdent_Int, selectStringExpr BIND
_locVarRecordIdent

SQL_PREPARE handleIdent_Int, retCodeIdent_Int, selectStringExpr BINDOUT
_locVar1, _locVar2, ...

SQL_PREPARE handleIdent_Int, retCodeIdent_Int, selectStringExpr BINDOUT
_locVarRowIdent

SQL_PREPARE handleIdent_Int, retCodeIdent_Int, selectStringExpr BINDOUT
_locVarRecordIdent
```

**Parametre**

handleIdent_Int	in	Identifikátor typu <i>Int</i> - jednoznačné číslo (handle) spojenia s databázou.
retCodeIdent_Int	out	Identifikátor typu <i>Int</i> - návratový kód.
selectStringExpr	in	Výraz typu <i>Text</i> .
_locVar1, _locVar2, ...	in	Zoznam lokálnych premenných.
_locVarRowIdent	in	Odkaz na riadok lokálnej premennej typu <i>Record</i> .
_locVarRecordIdent	in	Identifikátor lokálnej premennej typu <i>Record</i> .

**Návratový kód**

Hodnota parametra *retCodeIdent\_Int* - pozri tabuku [chybových stavov](#). Je možné získa [rozšírenú informáciu o chybe](#).

**Popis**

ítanie databázy SQL príkazom **SELECT** je implementované v dvoch alebo troch fázach. Prvú (prípravnú) fázu vykoná akcia **SQL\_PREPARE**. Nad databázou (je daná hodnotou identifikátora *handleIdent\_Int*) pripraví (a pokiaľ nie je použité kúové slovo **BINDOUT**, aj vykoná) príkaz **SELECT** definovaný hodnotou výrazu *selectStringExpr*. Úspešnosť akcie je indikovaná hodnotou *retCodeIdent\_Int*. Druhá fáza je potrebná vtedy, keď bolo použité kúové slovo **BINDOUT**. Znamená to, že vo výraze *selectStringExpr* bola použitá [parametrizácia](#) a je nutné volaním **SQL\_BINDIN** nastaviť hodnoty vstupných parametrov výrazu *selectStringExpr* a následne výraz vykona. Posledná fáza je postupné ítanie riadkov, ktoré pripravil príkaz **SELECT** akciou [SQL\\_FETCH](#).

ítanie prebieha zásadne do lokálnych premenných, ktoré sú uvedené za kúovým slovom **BIND** alebo **BINDOUT** akcie **SQL\_PREPARE**. Tu sú tri možné **varianty**:

- Zoznam neštruktúrovaných lokálnych premenných.**  
ítanie prebieha po jednom riadku do lokálnych premenných, ktoré sú vymenované za kúovým slovom **BIND** alebo **BINDOUT**.
- Odkaz na jeden riadok lokálnej premennej typu *Record*.**  
ítanie prebieha po jednom riadku do jedného riadku lokálnej štruktúrovanej premennej. Štruktúra ítaných dát musí byť zhodná so štruktúrou lokálnej premennej.
- Odkaz na lokálnu premennú typu *Record*.**  
ítanie prebieha po jednom alebo viacerých riadkoch lokálnej premennej. Tejto je podľa potreby pred naplnením výsledku ítania zmenený rozmer. Štruktúra ítaných dát musí byť zhodná so štruktúrou lokálnej premennej.

Pre jedno handle (získané akciou [SQL\\_CONNECT](#)) môže byť súčasne aktívne len jedno íťanie. Akcia [SQL\\_PREPARE](#) zruší platnosť predchádzajúcej akcie. Explicitne ukoní íťanie je možné pomocou akcie [SQL\\_FREE](#).

**Poznámka:** použitím [parametrizácie](#) je možné ušetriť prácu SQL databázy, pretože príprava (kompilácia) parametrizovaného SQL výrazu sa vykoná iba raz (v rámci akcie [SQL\\_PREPARE](#)). Následne je nutné nastaviť parametre akciou [SQL\\_BINDIN](#) (ktorá SQL príkaz aj vykoná) a jeden alebo viackrát volať [SQL\\_FETCH](#) na získanie výsledkov. Potom je možné nastaviť nové hodnoty parametrov a znovu vykonať SQL príkaz opätovným volaním [SQL\\_BINDIN](#) a získa nové výsledky volaním [SQL\\_FETCH](#). Vhodným nastavením parametrov databázy (napr. Oracle: *session\_cached\_cursors*) je možné zabezpečiť recyklovanie kurzorov (skompilovaných príkazov) medzi volaniami [SQL\\_PREPARE](#).

#### Príklad

[Príklad práce s databázou \(akcie SQL\\_ ...\)](#).

```

    BOOL _useBinding = @TRUE ; use parametrized SQL command
    INT  _handle      ; handle to database
    INT  _retCode     ; return code
    TEXT _name        ; product name
    TEXT _type        ; product type

                                ; parametrized SQL command
    TEXT _sqlPar = "SELECT Name, Type FROM Products WHERE ID>= #PAR# AND
ID<= #PAR#"

                                ; non-parametrized SQL command
    TEXT _sqlNpar = "SELECT Name, Type FROM Products WHERE ID>= 1 AND ID<=
100"

    SQL_CONNECT MyDatabase, _handle, _retCode

    IF _useBinding THEN ; parametrized alternative
        SQL_PREPARE _handle, _retCode, _sqlPar BINDOUT _name, _type
        SQL_BINDIN  _handle, _retCode, 1, 100 ; read all products between 1 and
100
    ELSE ; non-parametrized alternative
        SQL_PREPARE _handle, _retCode, _sqlNpar BIND _name, _type
    ENDIF

    DO_LOOP
        SQL_FETCH _handle, _retCode
        EXIT_LOOP _retCode # _ERR_NO_ERROR
        ; data processing goes here
    END_LOOP

    SQL_FREE _handle
    SQL_DISCONNECT _handle
```

#### Súvisiace odkazy

[DB\\_TRANS\\_OPEN](#)  
[DB\\_TRANS\\_COMMIT](#)  
[DB\\_TRANS\\_ROLLBACK](#)  
[DB\\_TRANS\\_CLOSE](#)

[SQL\\_CONNECT](#)  
[SQL\\_DISCONNECT](#)  
[SQL\\_EXEC\\_DIRECT](#)  
[SQL\\_EXEC\\_PROC](#)

[SQL\\_BINDIN](#)  
[SQL\\_FETCH](#)  
[SQL\\_FREE](#)

[SQL\\_SELECT](#)

[Všetky databázové akcie](#)



#### Súvisiace stránky:

[Akcie v skriptoch](#)

